# Is there a plot style for PolarPlot to look like matlab polar()?

The plot style of Matlab's polar function

appears to be pretty conventional for antenna radiation patterns in the references I'm looking at (sometimes with tick markings that count +- from the theta = 0 direction)

The basic plot can be done easily enough with Mathematica:

PolarPlot[ Cos[2 t] Sin[2 t], {t, 0, 2 Pi}]


but doesn't have the circular and arc scales, nor angular tick markings. Those probably wouldn't be hard to add with Epilog, but I was wondering if there's already an option to do this easily in Mathematica without writing a bunch of code?

If I understand the question correctly,

PolarPlot[Cos[2 t] Sin[2 t], {t, 0, 2 Pi},
PlotStyle -> Directive[Red, Dashed], PolarAxesOrigin -> {Pi/2, .5},
PolarGridLines -> Automatic, PolarAxes -> True,
PolarTicks -> {Table[{N[Pi/6 i], ToString[30 i] <> "°"}, {i, 0, 11}],
Table[{0.1 i, 0.1 i}, {i, 4}]}]


produces the desired results.

If "tick markings that count +- from the theta = 0 direction" are desired, replace the first Table in the code just above by

Table[{N[Pi/6 i], ToString[Abs[30 Mod[i, 12, -6]]] <> "°"}, {i, 0, 11}]